
Adding an Elevator 
By: Steven McMillan 
Well here is a way to script a very simple and basic lift that moves from point A to point B from a 
press of a button.  
 
For this to work you need to make script_object with targetname "elevator" (without quotes) and a 
trigger_use with targetname "elevator_button" (again no quotes) 
 
The elevator works as the moving elevator platform, and the elevator_button as a  a button, which 
player uses and activates the elevator.  
 
Remember tough that the trigger_button will be invisible, so make a button out of brushes.  
 
 
main:  
 
level waittill prespawn  
 
thread elevator_preparation //Set the threads...  
 
level waittill spawn  
 
end  
 
//----------------------------------------------------------->  
// Prepares the elvator by assigning the basic settings to it  
//----------------------------------------------------------->  
 
elevator_preparation:  
 
$elevator moveto $waypoint_down // Moves elevator to waypoint_down.  
$elevator_button bind $elevator // Attaches the elevator_button into the elevator, this making it 
move same time as the elevator.  
$elevator time 4 // sets the elevator move time, 4 seconds in this case, could also use the command 
"speed".  
local.elepos = 0 // Defines the position of the elevator. "0 - lift down" "1 - lift up" Default - lift down  
goto elevator_standby // standing by and waiting player to activate the trigger  
 
//----------------------------------------------------------->  
// Elevator is idle, and waiting user to activate the trigger  
//----------------------------------------------------------->  
 
elevator_standby:  
 
$elevator_button waittill trigger  
if (local.elepos == 0) // If lift state is down, its better get up.  
{  
println "elevator down - moving up" //prints this text to console if the developer mode is set.  
goto lift_move_up // jumps straight to the "lift_move_up" part in the script  
}  
else  



if (local.elepos == 1) // Or maybe lift state is up, and you need to get down?  
{  
println "elevator up - moving down" //prints this text to console if the developer mode is set.  
goto lift_move_down // jumps straight to the "lift_move_down" part in the script  
}  
 
//----------------------------------------------------------->  
 
lift_move_up:  
 
$elevator moveto $waypoint_up //moves the lift to info_waypoint with targetname waypoint_up  
$elevator playsound elevator_run //plays the "sound/mechanics/Mec_ElevatorRun_01.wav" defined 
in the ubersound.scr line 1052  
$elevator waitmove //wait till the elevator has completed the move before doing anything else  
local.elepos = 1 // it seems the elevator is up now, so we set the local.elepos to 1  
goto elevator_standby // Jumps back to the standby part of the script waiting user to activate the 
trigger  
 
//----------------------------------------------------------->  
 
lift_move_down:  
 
$elevator moveto $waypoint_down //moves the lift to info_waypoint with targetname 
waypoint_down  
$elevator playsound elevator_run //plays the "sound/mechanics/Mec_ElevatorRun_01.wav" defined 
in the ubersound.scr line 1052  
$elevator waitmove //wait till the elevator has completed the move before doing anything else  
local.elepos = 0 // it seems the elevator is down now, so we set the local.elepos to 1  
goto elevator_standby // Jumps back to the standby part of the script waiting user to activate the 
trigger  
 
//----------------------------------------------------------->  
 
end //the end. 


